
An Isabelle/HOL Formalisation of Green’s Theorem

Mohammad Abdulaziz† and Lawrence C. Paulson ‡

†Canberra Research Lab., NICTA, Australia?; †Australian National University; ‡ Computer
Laboratory, University of Cambridge, England

Abstract. We formalise a statement of Green’s theorem in Isabelle/HOL, which
is its first formalisation to our knowledge. The theorem statement that we for-
malise is enough for most applications, especially in physics and engineering. An
interesting aspect of our formalisation is that we neither formalise orientations
nor region boundaries explicitly, with respect to the outwards-pointing normal
vector. Instead we refer to equivalences between paths.

1 Introduction

The Fundamental Theorem of Calculus (FTC) is a theorem of immense importance in
differential calculus and its applications, relating a function’s differential to its integral.
Having been conceived in the seventeenth century in parallel to the development of
infinitesimal calculus, more general forms of the FTC have been developed, the most
general of which is referred to as the General Stokes’ Theorem

A generalisation of the FTC or a special case of the General Stokes’ Theorem in R2

was published in 1828 by George Green [2], with applications to electromagnetism in
mind. This generalisation is referred to as Green’s Theorem, and it is the main topic of
this work. In modern terms the theorem can be stated as follows:

Theorem 1. Given a domain D with an “appropriate” positively oriented boundary
∂D, and a field F , with components Fx and Fy “appropriately” defined on D, the
following identity holds:∫

D

∂Fy
∂x
− ∂Fx

∂y
dxdy =

∮
∂D

Fxdx+ Fydy,

where the left hand side is a double integral and the right hand side is a line integral1

in R2.

? NICTA is funded by the Australian Government through the Department of Communications
and the Australian Research Council through the ICT Centre of Excellence Program.

1 This line integral can be physically interpreted as the work done by F on the ∂D, making
this statement a special case of the 3-dimensional Kelvin-Stokes’ theorem. If the line integral
is replaced with

∮
∂D

Fxdx − Fydy, it can be interpreted as the flux of F through ∂D and the

theorem would be the 2-dimensional special case of the divergence theorem.



Many statements of Green’s theorem define with varying degrees of generality what
is an appropriate boundary (i.e. the geometrical assumptions), and what is an appropri-
ate field (i.e. the analytic assumptions). This mainly depends on how geometrically
sophisticated the proof is, and the underlying integral. The prevalent text book form of
Green’s theorem asserts that, geometrically, the region can be divided into elementary
regions and that, analytically, the field is continuous and has continuous partial deriva-
tives throughout the region. Also, usually the underlying integral is a Riemann integral.

Despite this being enough for most applications, more general forms of the theo-
rem have been proved in the analysis literature. Although a full literature review is not
appropriate here, we present some examples of very general formulations of Green’s
theorem. For example, J.H. Michael [7] proves a statement of the theorem that gener-
alises the geometrical assumptions, where it only assumes that the region has a rectifi-
able cycle as its boundary. Jurkat et al. [6] prove a statement of the theorem with very
general analytic assumptions on the field. They only assume that the field is continuous
in the region, and that the total derivative of the field exists in the region except for
a σ1-finite set of points in the region. Then, they also derive a very general form of
Cauchy’s integral theorem.

Having Green’s theorem formalised is significant because of its wide range of ap-
plications, too many to list in full. There are applications in physics (electrodynamics,
mechanics, etc.) and engineering (deriving moments of inertia, hydrodynamics, the ba-
sis of the planimeter, etc.). In mathematics, Green’s theorem is a fundamental result: it
can be used to derive Cauchy’s integral theorem.

We formalise a statement of Green’s theorem for Henstock-Kurtzweil gauge inte-
grals in the interactive theorem prover Isabelle/HOL [8]. Our work builds on the work
of Hölzl et.al., where we use the Isabelle/HOL multivariate analysis library [5] and the
probability library [4]. We also build on the second author’s porting of John Harrison’s
complex analysis library [3]. Our formalisation does not strictly follow a certain proof,
but it was inspired by Zorich and Cooke [11], Spivak [10] and Protter [9].

2 Basic Concepts and Lemmas

In this section we discuss the basic lemmas we need to prove Green’s theorem. How-
ever, we need to firstly discuss two basic definitions needed to state the theorem state-
ment: line integrals and partial derivatives. Definitions of both of those two concepts
are ubiquitous in the literature, nonetheless, we had to adapt them to be defined on the
Euclidean spaces type class in the Isabelle multivariate analysis library.

We define the line integral of a function F on the parameterised path γ as follows:

Definition 1. Line Integral∫
γ

F �B =

∫ 1

0

ΣbεB((F (γ(t)) · b)(γ′(t) · b))dt

A difference in our definition is that we add the argument B, a set of vectors, to
which F and γ, and accordingly the line integral are projected. The reasons for adding



the B argument will become evident later. Above, · denotes the inner product of two
vectors. Formally, the definition is:

definition line_integral::
"(’a::euclidean_space ⇒ ’a ) ⇒ ’a set ⇒ (real ⇒ ’a) ⇒ real"
where "line_integral F B γ =
integral {0 .. 1}
(λt.

∑
b∈B. (F(γ t) · b) *

(vector_derivative γ (at t within {0..1}) · b))"
definition line_integral_exists where
"line_integral_exists F B γ =

((λt.
∑

b∈B. F (γ t) · b *
(vector_derivative γ (at t within {0..1}) · b))

integrable_on {0..1})"

Note that integral refers to the Henstock-Kurzweil gauge integral implementation in
Isabelle/HOL library. As one would expect, the line integral distributes over unions of
sets of vectors and path joins.

lemma line_integral_sum_gen:
assumes "finite B"

"line_integral_exists F B1 γ"
"line_integral_exists F B2 γ"
"B1 ∪ B2 = B" "B1 ∩ B2 = {}"

shows "line_integral F B γ =
(line_integral F B1 γ) + (line_integral F B2 γ)"

lemma line_integral_distrib:
assumes "line_integral_exists f B γ1"

"line_integral_exists f B γ2"
"valid_path γ1" "valid_path γ2"

shows "line_integral f B (γ1 +++ γ2) =
line_integral f B γ1 + line_integral f B γ2"

The line integral also admits a transformation equivalent to integration by substitution.
This lemma applies to paths where all components are defined as a function in terms
of one component orthogonal to all of them. It is a critical lemma for proving Green’s
theorem for “elementary” regions (to be defined later).

lemma line_integral_on_pair_path:
fixes F::"(’a::euclidean_space) ⇒ (’a)"

g::"real ⇒ ’a"
γ::"(real ⇒ ’a)"
b::’a

assumes "b ∈ Basis"
"∀x. g(x) · b = 0"
"γ = (λt. f(t) *R b + g(f(t)))"
"γ C1_differentiable_on {0 .. 1}"
"continuous_on (f ‘ {0..1}) g"
"(pathstart γ) · b ≤ (pathfinish γ) · b"



"continuous_on (path_image γ) (λx. F x · b)"
shows "(line_integral F {b} γ)

= integral (cbox ((pathstart γ) · b)
((pathfinish γ) · b))

(λf_var. (F (f_var *R b + g(f_var)) · b))"

Partial derivatives are defined on the Euclidean space type class implemented in
Isabelle/HOL. For a function F defined on a Euclidean space, we define its partial
derivative to be w.r.t. the change in the magnitude of a component vector b of its input.
At a point a, the partial derivative is defined as:

Definition 2. Partial Derivative

∂F (v)

∂b

∣∣∣∣
v=a

=
dF (a+ (x− a · b)b)

dx

∣∣∣∣
x=a·b

Again, this definition is different from the classical definition in that the partial deriva-
tive is w.r.t. the change of magnitude of a vector rather than the change in one of the
variables on which F is defined. However, our definition is similar to the classical defi-
nition of a partial derivative, when b is a base vector. Formally we define it as:

definition has_partial_vector_derivative::
"(’a::euclidean_space ⇒ ’b::euclidean_space) ⇒ ’a ⇒ ’b ⇒ ’a
⇒ bool" where
"has_partial_vector_derivative F b F’ a
= ((λx. F( (a - ((a · b) *R b)) + x *R b ))

has_vector_derivative F’) (at (a · b))"

definition partially_vector_differentiable where
"partially_vector_differentiable F b p

= (∃F’. has_partial_vector_derivative F b F’ p)"

definition partial_vector_derivative::
"(’a::euclidean_space ⇒ ’b::euclidean_space) ⇒ ’a ⇒ ’a ⇒ ’b"
where "partial_vector_derivative F b a

= (vector_derivative
(λx. F( (a - ((a · b) *R b)) + x *R b)) (at (a · b)))"

The following FTC for the partial derivative follows from the FTC for the vector deriva-
tive that is proven in Isabelle/HOL analysis library.

lemma fundamental_theorem_of_calculus_partial_vector_gen:
fixes k1 k2::"real"

F::"(’a::euclidean_space ⇒ ’b::euclidean_space)"
b::"’a"
F_b::"(’a::euclidean_space ⇒ ’b)"

assumes "k1 ≤ k2"
"b · b = 1"
"c · b = 0 "
"∀ p ∈ D. has_partial_vector_derivative F b (F_b p) p"
"{v. ∃x. k1 ≤ x ∧ x ≤ k2 ∧ v = x *R b + c} ⊆ D"



shows "( (λx. F_b( x *R b + c)) has_integral
F(k2 *R b + c) - F(k1 *R b + c)) (cbox k1 k2)"

Given these definitions and basic lemmas, we can now start elaborating on our for-
malisation of Green’s theorem. The first issue is how to formalise R2. We use pairs of
real to refer to members of R2, where we define the following locale that fixes the
base vector names:

locale R2 =
fixes i j
assumes i_is_x_axis: "i = (1::real,0::real)" and

j_is_y_axis: "j = (0::real, 1::real)"

Proofs of Green’s theorem usually start by proving “half” of the theorem statement
for every type of “elementary regions” in R2. These regions are referred to as Type I,
Type II or Type III regions, defined below.2

Definition 3. Elementary Regions
A region D (modelled as a set of real pairs) is Type I iff there are C1 smooth functions
g1 and g2 such that for two constants a and b:

D = {(x, y) | a ≤ x ≤ b ∧ g2(x) ≤ y ≤ g1(x)}.

Similarly D would be called type II iff for g1, g2, a and b

D = {(x, y) | a ≤ y ≤ b ∧ g2(y) ≤ x ≤ g1(y)}.

Finally, a region is of type III if it is both of type I and type II.

To prove Green’s theorem a common approach is to prove the following two “half”
Green’s theorems, for any regions Dx and Dy that are type I and type II, respectively,
and their positively oriented boundaries:∫

Dx

− ∂(Fi)

∂j
dxdy =

∫
∂Dx

F �{i},

and ∫
Dy

∂(Fj)

∂i
dxdy =

∫
∂Dy

F �{j}.

Here i and j are the base vectors while Fi and Fj are the x-axis and y-axis components,
respectively, of F . However, the difference in the expressions for the type I and type II
regions is because of the asymmetry of the x-axis and the y-axis w.r.t. the orientation.
We refer to the top expression as the x-axis Green’s theorem, and the bottom one as the
y-axis Green’s theorem. Below is the statement of the x-axis Green’s theorem for type I
regions as we have formalised it in Isabelle/HOL. For the boundary, we model its paths
explicitly as functions of type real ⇒ (real * real), where γ1, γ2, γ3 and γ4
are the bottom, right, top and left sides, respectively.

2 Using elementary regions that are bounded by C1 smooth functions is as general as using
piece-wise smooth functions because it can be shown that the latter can be divided into regions
of the former type (see [9]).



lemmaGreens_thm_type_I:
fixes F::"((real*real) ⇒ (real * real))"

γ1 γ2 γ3 γ4 ::"(real ⇒ (real * real))"
a::"real" b::"real"
g1::"(real ⇒ real)" g2::"(real ⇒ real)"

assumes "Dx = {(x,y) . x ∈ cbox a b ∧ y ∈ cbox (g2 x) (g1 x)}"
"γ1 = (λt. (a + (b - a) * t, g2(a + (b - a) * t)))"
"γ1 C1_differentiable_on {0..1}"
"γ2 = (λt. (b, g2(b) + t *R (g1(b) - g2(b))))"
"γ3 = (λt. (a + (b - a) * t, g1(a + (b - a) * t)))"
"γ3 C1_differentiable_on {0..1}"
"γ4 = (λt. (a, g2(a) + t *R (g1(a) - g2(a))))"
"analytically_valid Dx (λp. F(p) · i) j"
"∀x ∈ cbox a b. (g2 x) ≤ (g1 x)"
"a < b"

shows "(line_integral F {i} γ1) + (line_integral F {i} γ2) -
(line_integral F {i} γ3) - (line_integral F {i} γ4) =
(integral Dx
(λa. - (partial_vector_derivative (λp. F(p) · i) j a)))"

Proving the lemma above depends on the observation that for a path γ (e.g. γ1 above)
that is straight along a vector x (e.g. i),

∫
γ

F �{x} = 0, for an F continuous on γ.

(Formally, this observation follows immediately from theorem work_on_pair_path.)
The rest of the proof boils down to an application of Fubini’s theorem and the FTC to
the double integral, the integral by substitution to the line integrals and some algebraic
manipulation ([11, page 238]). Nonetheless, this algebraic manipulation proved to be
quite tedious when done formally in Isabelle/HOL.

However, we did not discuss the predicate analytically_valid, which repre-
sents the analytic assumptions of our statement of Green’s theorem, to which an “ap-
propriate” field has to conform. Firstly let 1s be the indicator function for a set s. Then,
for the x-axis Green’s theorem, our analytic assumptions are that

(i) Fi is continuous on Dx

(ii) ∂(Fi)
∂j exists everywhere in Dx

(iii) the product 1Dx
(x, y)∂(Fi)

∂j (x, y) is Lebesgue integrable

(iv) the product 1[a,b](x)
∫ g2(x)
g1(x)

F (x, y)dy, where the integral is a Henstock-Kurzweil
gauge integral, is a borel measurable function

These assumptions vary symmetrically for the y-axis Green’s theorem, so to avoid hav-
ing two symmetrical definitions, we defined the predicate analytically_valid to
take the axis (as a base vector) as an argument.

definition analytically_valid::
"’a::euclidean_space set ⇒
(’a ⇒ ’b::{euclidean_space, times, one, zero}) ⇒ ’a ⇒ bool"

where
"analytically_valid s F b =
((∀ a ∈ s. partially_vector_differentiable F b a) ∧



continuous_on s F ∧
integrable lborel
(λp. (partial_vector_derivative F i) p * indicator s p) ∧

(λx. integral UNIV
(λy. (partial_vector_derivative F i

(y *R b + x *R (
∑

b ∈(Basis - {i}). b))) *
(indicator s

(y *R b + x *R (
∑

b ∈(Basis - {i}). b)) )))
∈ borel_measurable lborel)"

These conditions refer to Lebesgue integrability and to measurability because we
use Fubini’s theorem for the Lebesgue integral in Isabelle/HOL’s probability library to
derive a Fubini like result for the Henstock-Kurzweil integral. Proving Fubini’s theorem
for the gauge integral would allow for more general analytic assumptions. However, the
rest of our approach would still be valid.

We prove the y-axis Green’s theorem for type II regions similarly, where this is its
conclusion.

shows "(line_integral F {j} γ1) + (line_integral F {j} γ2) -
(line_integral F {j} γ3) - (line_integral F {j} γ4) =
(integral Dy
(partial_vector_derivative (λp. F(p) · j) i))"

3 More General Structures

Now that we described some of the basic definitions and how to derive Green’s theorem
for elementary regions, the remaining question is how to prove the theorem for more
general regions. As we stated earlier in the introduction, a lot of the text book proofs of
Green’s theorem are given for regions that can be divided into elementary regions. It can
be shown that any regular region can be divided into elementary regions [9,11]. Regular
regions (see their definition in [9, page 235]), are enough for a lot of applications,
especially practical applications in physics and engineering.

In this section we describe how we prove Green’s theorem for regions that can be
divided both into type I regions and type II regions only using vertical and horizontal
edges, respectively. We believe that for most practical purposes, the additional assump-
tion that the division is done only by vertical and horizontal edges is equivalent to
assuming just the existence of type I and type II divisions. Indeed, we conjecture that
the additional constraints do not lead to any loss of generality, however, we would not
pursue the proof of this claim in the current paper.

Figure 1 shows an example of a region and its type I and type II partitions. In this
example, some of the elementary regions appear to have a missing edge. This is because
the type I or the type II partitioning induced a one-point path: a function mapping the
interval [0, 1] to a single point in R2. For instance, the left edge in the left 1-chain in 1b
is a point on the x-axis.



(a) (b) (c) (d)

(e) (f) (g) (h)

(i) (j)

Fig. 1: An annulus and its partitioning in type I and type II regions. In this figure every
1-cube (i.e. path) is represented with an arrow whose direction is the same as the ori-
entation of the 1-cube. a) The positively oriented boundary of the annulus. b), c) and
d) The members of a type I partitioning of the annulus. e) A 1-chain that includes all
the horizontal boundaries in the type I partition. f), g) and h) The members of a type II
partitioning of the annulus. i) A 1-chain that includes all the vertical boundaries in the
type II partition. j) A common subdivision of the chains in e) and i).



3.1 Chains and cubes

For tackling more general regions and their boundaries we use the concepts of cubes and
chains [10, chapter 8]. One use of cubes is to represent parameterisable (sub)surfaces
(regions in R2 and paths in our case). A k-dimensional such surface embedded in Rn
is represented by a function whose domain is a space homeomorphic to Rk and whose
codomain is Rn. Roughly speaking, we model cubes as functions and chains as sets
of cubes. We use the existing Isabelle/HOL formalisation of paths, where we model
1-cubes as functions defined on the interval {0..1}. We model a 1-chain as a set of
pairs of int (coefficients) and 1-cubes. For example, the following definition shows the
lifting of the line integrals to 1-chains.

definition one_chain_line_integral ::
"((real*real) => (real *real)) => ((real*real) set) ⇒
(int * (real ⇒real*real)) set ⇒ real" where

"one_chain_line_integral F b C =
(
∑

(k,γ)∈C. k * (line_integral F b γ))"

We extend the way we model 1-cubes to model 2-cubes, which we model as functions of
type (real * real ⇒ real * real) defined on the interval {(0,0)..(1,1)}.

definition cubeImage::
"(real*real ⇒ real*real) ⇒ ((real*real) set)" where

"cubeImage twoC = (twoC ‘ (cbox (0,0) (1,1)))"

The orientation of the boundary of a 2-cube (a 1-chain) is taken to be counter-clockwise.
A 1-cube is given the coefficient −1 if the path’s direction is against the counter-
clockwise boundary traversal order, otherwise it is given the coefficient 1. Formally
this is defined as follows:

definition horizontal_boundary ::
"(real*real ⇒ real*real) ⇒ (real* (real ⇒ real*real)) set"
where "horizontal_boundary twoC =

{(1, (λx. twoC(x,0))), (-1, (λx. twoC(x,1)))}"

definition vertical_boundary ::
"(real*real ⇒ real*real) ⇒ (real* (real ⇒ real*real)) set"
where "vertical_boundary twoC =

{(-1, (λy. twoC(0,y))), (1, (λy. twoC(1,y)))}"

definition boundary where
"boundary twoC =

horizontal_boundary twoC ∪ vertical_boundary twoC"

We follow the convention and define the 2-cubes in such a way that the top and left
edges are against the counter-clockwise orientation (e.g. see the 2-cube in Figure 1c).
Accordingly both the left and top edges take a −1 coefficient in the 1-cube representa-
tion. Defining 2-cubes in that way makes it easier to define predicates identifying type
I and type II 2-cubes, as follows.



definition typeI_twoCube ::
"((real * real) ⇒ (real * real)) ⇒ bool" where
"typeI_twoCube twoC ←→
(∃ (a::real) (b::real) (g1::real⇒real) (g2::real ⇒ real).
a < b ∧ (∀x ∈ {a..b}. g2 x ≤ g1 x) ∧
twoC =
(λ(t1,t2). ((1-t1)*a + t1*b,

(1 - t2) * (g2 ((1-t1)*a + x*b)) +
t2 * (g1 ((1-t1)*a + t1*b))))∧

g1 C1_differentiable_on {a .. b} ∧
g2 C1_differentiable_on {a .. b})"

Although we do not render it here, an analogous predicate is defined for type II 2-cubes.
We also require that all 2-cubes conform to the following predicate:

definition valid_two_cube where
"valid_two_cube twoC = (card (boundary twoC) = 4)"

This predicate filters out cases where 2-cubes have either: i) right and top edges that
are both one point paths, or ii) left and bottom edges that are both one point paths.
Although this assumption potentially leaves our theorems less general regarding some
corner cases, it makes our computations much smoother. After defining these concepts
on 2-cubes, we derive the following statement of x-axis Green’s theorem (and its equiv-
alent y-axis Green’s theorem) in terms of 2-cubes.

GreenThm_typeI_twoCube:
assumes "typeI_twoCube twoC"
"valid_two_cube twoC"
"analytically_valid (cubeImage twoC) (λx. (F x) · i) j"

shows "integral (cubeImage twoC)
(λa. - partial_vector_derivative (λp. F p · i) j a)
= one_chain_line_integral F {i} (boundary twoC)"

Although we anticipated that proving this theorem would be a straightforward unfolding
of definitions and usage of GreenThm_typeI, it was a surprisingly long and tedious
proof that took a few hundred lines.

For 2-chains, we model them as sets of 2-cubes, which suits our needs in working
in the context of Green’s theorem. We define the boundary of a 2-chain as follows:

definition two_chain_boundary::
"((((real*real) ⇒ (real*real))) set) ⇒
((real* (real ⇒ (real*real))) set)" where

"two_chain_boundary twoChain =
⋃
(boundary ‘ twoChain)"

We similarly defined the functions two_chain_horizontal_boundary and two_

chain_vertical_boundary. We also lift the double integral to 2-chains as follows.

definition two_chain_integral::
"((((real*real) ⇒ (real*real))) set) ⇒
((real*real)⇒(real)) ⇒ real" where

"two_chain_integral twoChain F =
(
∑

(C)∈twoChain. (integral (cubeImage C) F))"



Lastly, to smoothe our computations on integrals over 2-chains and their boundaries, we
require that a 2-chain: i) only has valid 2-cubes members, ii) edges of different 2-cubes
only coincide if they have opposite orientations, and iii) different 2-cubes have different
images. These requirements are formally defined in the following predicate:

definition valid_two_chain where
"valid_two_chain twoChain =
((∀twoCube ∈ twoChain. valid_two_cube twoCube) ∧
(∀twoCube1 ∈ twoChain.
∀twoCube2 ∈ twoChain.

twoCube1 6= twoCube2 −→
boundary twoCube1 ∩ boundary twoCube2 = {}) ∧

inj_on cubeImage twoChain)"

Given these definitions on 2-chains, we lift our x-axis Green’s theorem from 2-
cubes to 2-chains, as shown in the following statement.

lemma GreenThm_typeI_twoChain:
assumes "∀twoCube ∈ twoChain. typeI_twoCube twoCube"

"valid_two_chain twoChain"
"∀twoC ∈ twoChain.

analytically_valid (cubeImage twoC) (λx. (F x) · i) j"
shows "two_chain_integral twoChain

(λa. - partial_vector_derivative
(λx. (F x) · i) j a) =

one_chain_line_integral F {i}
(two_chain_boundary twoChain)"

A similar lemma can be derived for a 2-chain whose members are all of Type II (with
the obvious consequences of orientation asymmetry).

After proving the x-axis and y-axis Green’s theorems, the next step is, after alge-
braic and analytic manipulation, to add the line integral sides of the x-axis Green’s
theorem to its counterpart in the y-axis theorem and similarly add the double integrals
of both theorems. Given GreenThm_typeI_twoChain and its type II equivalent, we
can sum up both sides of the equalities in the conclusion and get Green’s theorem in
terms of 2-chains and their boundaries. However, the main goal of the paper is to obtain
the theorem directly for a region and its boundary, given that the region can be vertically
sliced intro regions of type I and horizontally sliced into regions of type II.

The first (and easier) part in proving this is to prove the equivalence of the double
integral on a region and the integral on a 2-chain that divides that region. Before deriving
such a theorem we generalised the notion of division_of, defined in Isabelle/HOL’s
multivariate analysis library, to work when the division is not constituted of rectangles.

definition gen_division where
"gen_division s S =

(finite S ∧ (
⋃
S = s) ∧

(∀u ∈ S. ∀t ∈ S. u 6= t −→ negligible (u ∩ t)))"

Then we show the following equivalence:



lemma two_chain_integral_eq_integral_divisable:
assumes "∀ twoCube ∈ twoChain. F integrable_on cubeImage twoCube"

"gen_division s (cubeImage ‘ twoChain)"
"valid_two_chain twoChain"

shows "integral s F = two_chain_integral twoChain F"

The other part concerning the line integrals, proved to be trickier, and we will ex-
plain it in the next section.

3.2 Dealing with Boundaries

What remains now is to prove an equivalence between the line integral on the 1-chain
boundary of the region under consideration and the line integral on the 1-chain boundary
of the region’s elementary division (i.e. the 2-chain division of the region). The classical
approach depends on that the line integrals on the introduced boundaries will cancel
each other, leaving out the line integral on the region’s original boundary. For example,
the vertical-straight-line paths in Figures 1b, 1c and 1d, are the introduced boundaries
to obtain the type I division of the annulus. In this example, the line integrals on the
introduced vertical-straight-line paths will cancel each other because of their opposite
orientations, thus, leaving out the integral on the original boundary.

To prove this formally, the classical approach needs to define what is a positively
oriented boundary, which requires an explicit definition of the boundary of a region,
and also defining the exterior normal of the region (examples of this approach can be
found in [9,1]). However, we use a different approach that does not depend on these
definitions and avoids a lot of the resulting geometrical and analytic complications. Our
approach depends on two observations:

O1: if a path γ is straight along a vector x, then
∫
γ

F �{x} = 0, for an F continuous on γ.

O2: partitioning the region in type I/type II regions was done by introducing only verti-
cal/horizontal boundaries.

For a type I 2-chain division of a region, consider the 1-chain γx, that: i) includes all
the horizontal boundaries of the dividing 2-chain, and ii) includes some subpaths of the
vertical boundaries of the dividing 2-chain (call this condition Cx). Based on O1, the
line integral on the vertical edges in the 1-chain boundary of the type I division and ac-
cordingly γx, projected on i, will be zero. Accordingly we can prove the x-axis Green’s
theorem for γx. Formal statements of Cx (formally: only_vertical_division), and
the consequence of a 1-chain conforming to it are:

definition only_vertical_division where
"only_vertical_division one_chain two_chain =
(∃V. finite V ∧

(∀(k, γ) ∈ V.
(∃ (k’, γ’) ∈ two_chain_vertical_boundary two_chain.
(∃ a ∈ {0 .. 1}.
∃ b ∈ {0..1}.
a ≤ b ∧ subpath a b γ’ = γ))) ∧

one_chain = V ∪ (two_chain_horizontal_boundary two_chain))"



lemma GreenThm_typeI_divisible_region_boundary_gen:
assumes "valid_typeI_division s twoChain"

"∀twoC ∈ twoChain.
analytically_valid (cubeImage twoC) (λa. F(a) · i) j"

"only_vertical_division γ_x twoChain"
shows "integral s

(λx. - partial_vector_derivative (λa. F(a) · i) j x)
= one_chain_line_integral F {i} γ_x "

In the previous lemma, valid_typeI_division is an abbreviation that, for a region
and a 2-chain, means that the 2-chain constitutes only valid type I cubes and that this
2-chain is a division of the given region.

An analogous condition, Cy, for a type II partitioning asserts that the 1-chain in-
cludes all the vertical boundaries of the dividing 2-chain, and includes some subpaths
of the horizontal boundaries of the dividing 2-chain. For Cy, we formally define the
predicate (only_horizontal_division) and prove an analogous theorem for type
II partitions, with the obvious changes in the conclusion.

From the second observation, O2, we can conclude that there will always be

– a 1-chain, γx, whose image is the boundary of the region under consideration and
that satisfies Cx for the type I division.

– a 1-chain, γy , whose image is the boundary of the region under consideration and
that satisfies Cy for the type II division, where it is not necessary that γx = γy .

Figure 1e and Figure 1i show two 1-chains that satisfy Cx and Cy for the type I and
type II divisions of the annulus. Notice that in this example, those two 1-chains are not
equal even though they have the same orientation and image.

Now, if we can state and formalise the equivalence between γx and γy , and that this
equivalence lifts to equal line integrals, we can obtain Green’s theorem in terms of the
region, which is our goal. One way to formalise path equivalence is to explicitly define
the notion of orientation. Then the equivalence between γx and γy can be characterised
by their having similar orientations and images. An advantage of this approach is that
it can capture equivalence in path orientations regardless of the path image.

However, we do not need this generality in the context of proving the equivalence
of 1-chains that have the same image and orientation, especially that this generality will
cost a lot of analytic and geometric complexities to be formalised. Instead we choose
to foramlise the notion of equivalence in terms of having a common subdivision. For
example the 1-chain shown in Figure 1j is a subdivision of each of the 1-chains in
Figure 1e and Figure 1i as well as the original boundary 1-chain in Figure 1a, i.e. a
common subdivision between the three 1-chains. We now formally define the concept
of a common subdivision between 1-chains, where we mainly focus on “boundary”
1-chains, defined as follows.

definition boundary_chain where
"boundary_chain s = (∀(k, γ) ∈ s. k = 1 ∨ k = -1)"

First, we lift the path_join operator defined in the Isabelle/HOL multivariate anal-
ysis library, to act on 1-chains ordered into lists as follows.



fun rec_join ::
"(int *(real ⇒ (real*real))) list ⇒ (real ⇒ (real * real))"
where
"rec_join [] = (λx.0)" |
"rec_join ((k, γ) # xs) =

(if (k = 1) then γ else reversepath γ) +++ (rec_join xs)"

To use the theory for paths developed in the multivariate analysis library, we need the
joined chains to be piece-wise C1 smooth in the sense that is defined in that library
(valid_path). A necessary condition for a path to be valid, is that the ending point of
every piece of that path to be the starting point of the next. Accordingly we define the
following predicate for the validity 1-chains ordered into lists.

fun valid_chain_list where
"valid_chain_list [] = True" |
"valid_chain_list ((k, γ)#l) =
(( if (k = 1) then (pathfinish γ = pathstart (rec_join l))

else (pathfinish (reversepath γ) = pathstart (rec_join l)))
∧ valid_chain_list l)"

Based on those concepts we now define what it means for a 1-chain to be a subdivision
of a path, which is a straightforward definition.

definition chain_subdiv_path::
"(real ⇒ real * real) ⇒ ((int * (real ⇒ real * real)) set)
⇒ bool" where
"chain_subdiv_path γ one_chain

= (∃l. set l = one_chain ∧ distinct l ∧ rec_join l = γ ∧
valid_chain_list l)"

We call a 1-chain γ, a subdivision of another 1-chain η, if one can map every cube in η
to a sub-chain of γ that is a subdivision of it. Formally this is defined as follows:

definition chain_subdiv_chain where
"chain_subdiv_chain one_chain1 subdiv =
(∃f. ((

⋃
(f ‘ one_chain1)) = subdiv) ∧

(∀(k,γ)∈one_chain1.
if k = 1 then
chain_subdiv_path γ (f(k, γ))

else
chain_subdiv_path (reversepath γ) (f(k,γ))) ∧

(∀p∈one_chain1.
∀p’∈one_chain1. p 6= p’ −→ f p ∩ f p’ = {}) ∧

(∀x ∈ one_chain1. finite (f x)))"

After proving that each of the previous notions of equivalence implies equality of line
integrals, we define equivalence of 1-chains in terms of having a common subdivision,
and prove that it implies equal line integrals. We define it as having a boundary 1-
chain that is a subdivision for each of the 1-chains under consideration. Formally this
definition and the equality of line integrals that it implies are as follows:



definition common_boundary_sudivision_exists where
"common_boundary_sudivision_exists one_chain1 one_chain2 =

(∃subdiv. chain_subdiv_chain one_chain1 subdiv ∧
chain_subdiv_chain one_chain2 subdiv ∧
(∀(k, γ) ∈ subdiv. valid_path γ)∧
(boundary_chain subdiv))"

lemma common_subdivision_imp_eq_line_integral:
assumes "(common_boundary_sudivision_exists

one_chain1 one_chain2)"
"boundary_chain one_chain1"
"boundary_chain one_chain2"
"∀(k, γ)∈one_chain1. line_integral_exists F B γ"
"finite one_chain1"
"finite one_chain2"

shows "one_chain_line_integral F B one_chain1 =
one_chain_line_integral F B one_chain2"

"∀(k, γ)∈one_chain2. line_integral_exists F B γ"

Based on this lemma, finally, we prove the following statement of Green’s theorem.

lemma GreenThm_typeI_typeII_divisible_region:
assumes "valid_typeI_division s twoChain_typeI"

"valid_typeII_division s twoChain_typeII"
"∀twoC ∈ twoChain_typeI.

analytically_valid (cubeImage twoC) (λx. (F x) · i) j"
"∀ twoC ∈ twoChain_typeII.

analytically_valid (cubeImage twoC) (λx. (F x) · j) i"
"only_vertical_division γ_x twoChain_typeI"
"boundary_chain γ_x"
"only_horizontal_division γ_y twoChain_typeII"
"boundary_chain γ_y"
"common_boundary_sudivision_exists γ_x γ_y"

shows "integral s
(λx. partial_vector_derivative (λx. (F x) · j) i x -

partial_vector_derivative (λx. (F x) · i) j x)
= one_chain_line_integral F {i, j} γ_x"

"integral s
(λx. partial_vector_derivative (λx. (F x) · j) i x -

partial_vector_derivative (λx. (F x) · i) j x)
= one_chain_line_integral F {i, j} γ_y"

This theorem does not require the 1-chains γ_x and γ_y to have as their image exactly
the boundary of the region. However, of course it applies to the 1-chains if their image
is the boundary of the region. Accordingly it fits as Green’s theorem for a region that
can be divided into elementary regions just by vertical and horizontal slicing.

It is worth noting that although this statement seems to have a lot of assumptions,
its analytic assumptions regarding the field are strictly more general than the those in
[11,9], where they retuire the field and both of its partial derivatives to be continuous
in the region. For the geometric assumptions, on the other hand, we have two extra



requirements: the type I and type II divisions should be obtained using only vertical
slicing and only horizontal slicing, respectively.

4 Conclusion and Future Work

We formalised a statement of Green’s theorem that is enough for a lot of practical appli-
cations. Theory and concepts we developed here can be used in proving more general
statements of Green’s theorem [6,7]. Most such proofs depend on approximating both
line and double integrals on a region by corresponding integrals on a region that can be
divided into elementary regions. An interesting aspect of our work is that we avoided
defining the region’s boundary and its orientation explicitly. We did so by assuming
that the division was done by inserting only vertical edges for the type I division, and
only horizontal edges for the type II division. We claim that this added condition on the
division represents no loss of generality, and intend to prove this claim in the future.

Isabelle Notation and Availability All blocks starting with isabelle keywords: lemma,
definition, fun have been generated automatically using Isabelle/HOL’s LATEX pretty-
printing utility. Sometimes we have edited them slightly to improve readability, but the
full sources are available online.3

